Target Mass Corrections

All matrix elements use even integers and are equivalent to Dong' s a_{n+1}, d_{n+1}

From Bluemlein and Tkabladze's paper on TMCs

In[1]:=
```
Comb[n_, j_] := (y^2)^j (n + j + 1)! / (j! n! (n + 2 j + 1)^2)
Comb1[n_, j_] := (y^2)^j ((n + j + 1)! / (j! n! (n + 2 j + 1) (n + 2 j + 2)^2))
```

CN Moments

In[3]:=
```
g1tw2[n_, jmax_] := (n + 1) Sum[Comb[n, j] a[n + 2 j], {j, 0, jmax}]
g2tw2[n_, jmax_] := - (n) Sum[Comb[n, j] a[n + 2 j], {j, 0, jmax}]
g1tw3[n_, jmax_] := 4 Sum[Comb[n, j] j d[n + 2 j], {j, 0, jmax}]
g2tw3[n_, jmax_] := Sum[Comb1[n - 1, j] (n + 2 j) (n + 1) d[n + 2 j], {j, 0, jmax}]
g1[n_, jmax_] := g1tw2[n, jmax] + g1tw3[n, jmax]
g2[n_, jmax_] := If[n == 0, 0, g2tw2[n, jmax] + g2tw3[n, jmax]]
```

In[9]:=
```
g1[2, 8] + g2[2, 8] // Simplify
```

Out[9]=

$$\frac{a[2]}{3} + \frac{2\,d[2]}{3} + \frac{6}{25} y^2 (2\,a[4] + 11\,d[4]) + \frac{6}{49} y^4 (5\,a[6] + 46\,d[6]) + \frac{20}{27} y^6 (a[8] + 13\,d[8]) + \frac{15}{121} y^8 (7\,a[10] + 118\,d[10]) + \frac{42}{169} y^{10} (4\,a[12] + 83\,d[12]) + \frac{28}{75} y^{12} (3\,a[14] + 74\,d[14]) + \frac{72}{289} y^{14} (5\,a[16] + 143\,d[16]) + \frac{45}{361} y^{16} (11\,a[18] + 358\,d[18])$$

In[10]:=
```
2 g1[2, 8] + 3 g2[2, 8] // Simplify
```

Out[10]=

$$2 \left(d[2] + 3\,y^2\,d[4] + 6\,y^4\,d[6] + 10\,y^6\,d[8] + 15\,y^8\,d[10] + 21\,y^{10}\,d[12] + 28\,y^{12}\,d[14] + 36\,y^{14}\,d[16] + 45\,y^{16}\,d[18]\right)$$

Note the equation above is not correct in Dong' s paper.

In[11]:=
```
g2tw3[2, 3]
```

Out[11]=

$$\frac{2\,d[2]}{3} + \frac{18}{25} y^2\,d[4] + \frac{36}{49} y^4\,d[6] + \frac{20}{27} y^6\,d[8]$$

Nachtmann Moments

Definitions

In[12]:= $\xi = \frac{2\,x}{1+\sqrt{1+y^2\,4\,x^2}}$;

$\gamma = \sqrt{1+y^2\,4\,x^2}$;

$\rho = \sqrt{1+\gamma^2}$;

Solve[ξ == 1, x]

Out[15]= $\left\{\left\{x \to \frac{1}{1-y^2}\right\}\right\}$

Checking the Nachtmann moments

In[16]:= $\text{M1}[n\_] := \frac{\xi^{n+1}}{x^2}\left(\left(\frac{x}{\xi} - \frac{n^2}{(n+2)^2}\,y^2\,x\,\xi\right)\text{g1}[x] - y^2\,x^2\,\frac{4\,n}{n+2}\,\text{g2}[x]\right)$

$\text{M2}[n\_] := \frac{\xi^{n+1}}{x^2}\left(\frac{x}{\xi}\,\text{g1}[x] + \left(\frac{n}{n-1}\,\frac{x^2}{\xi^2} - \frac{n}{n+1}\,y^2\,x^2\right)\text{g2}[x]\right)$

$\text{Id2} := x^2\,(2\,\text{g1}[x] + 3\,\text{g2}[x])$

Expanding the moments in powers of y

```
In[19]:= nMax = 12;
Series[M1[n], {y, 0, nMax}] // Simplify;
M23integrand = Series[M2[3], {y, 0, nMax}] // Simplify // Normal;
Iintegrand = Series[Id2, {y, 0, nMax}] // Simplify // Normal;
M23parts = CoefficientList[M23integrand, x];
Iparts = CoefficientList[Iintegrand, x];
```

Note the factor of 2 below

In[25]:=
```
res =
  MapIndexed[#1 /. {g1[x] → g1[First[#2 - 1], nMax], g2[x] → g2[First[#2 - 1], nMax]} &,
   M23parts] // Simplify;
2 * Total[res] // Simplify

res =
  MapIndexed[#1 /. {g1[x] → g1[First[#2 - 1], nMax], g2[x] → g2[First[#2 - 1], nMax]} &,
   Iparts] // Simplify;
Total[res] // Simplify
```

Out[26]= $2\,(d[2] + 2\,y^{14}\,(1716\,d[16] + 23\,205\,y^2\,d[18] + 170\,170\,y^4\,d[20] + 895\,356\,y^6\,d[22] + 3\,779\,100\,y^8\,d[24] + 13\,579\,566\,y^{10}\,d[26] + 43\,088\,955\,y^{12}\,d[28] + 123\,735\,690\,y^{14}\,d[30] + 327\,085\,304\,y^{16}\,d[32] + 809\,924\,115\,y^{18}\,d[34] + 1\,829\,361\,534\,y^{20}\,d[36] + 4\,833\,678\,850\,y^{22}\,d[38]))$

Out[28]= $2\,(d[2] + 3\,y^2\,d[4] + 6\,y^4\,d[6] + 10\,y^6\,d[8] + 15\,y^8\,d[10] + 21\,y^{10}\,d[12] + 28\,y^{12}\,d[14] + 36\,y^{14}\,d[16] + 45\,y^{16}\,d[18] + 55\,y^{18}\,d[20] + 66\,y^{20}\,d[22] + 78\,y^{22}\,d[24] + 91\,y^{24}\,d[26])$

The first result shows that $M_2{}^3$ does indeed cleanly project out d_2 in the presence of target mass effects.

In[29]:=
```
Series[M1[3], {y, 0, 3}] // Simplify
Series[M2[3], {y, 0, 3}] // Simplify
```

Out[29]= $x^2\,g1[x] - \frac{12}{25}\,(x^4\,(7\,g1[x] + 5\,g2[x]))\,y^2 + O[y]^4$

Out[30]= $\frac{1}{2}\,x^2\,(2\,g1[x] + 3\,g2[x]) - \frac{3}{4}\,(x^4\,(4\,g1[x] + 5\,g2[x]))\,y^2 + O[y]^4$

Check for d' s

d2

In[31]:=
```
nMax = 12;
M11integrand = Series[M2[3], {y, 0, 8}] // Simplify // Normal;
M11parts = CoefficientList[M11integrand, x];
res =
  MapIndexed[#1 /. {g1[x] → g1[First[#2 - 1], nMax], g2[x] → g2[First[#2 - 1], nMax]} &,
   M11parts] // Simplify;
Total[res] // Simplify
```

Out[35]= $d[2] + y^{10}\,(297\,d[12] + 2860\,y^2\,d[14] + 15\,444\,y^4\,d[16] + 61\,425\,y^6\,d[18] + 200\,200\,y^8\,d[20] + 565\,488\,y^{10}\,d[22] + 1\,432\,080\,y^{12}\,d[24] + 3\,325\,608\,y^{14}\,d[26] + 7\,194\,720\,y^{16}\,d[28] + 14\,678\,820\,y^{18}\,d[30] + 28\,217\,280\,y^{20}\,d[32] + 58\,198\,140\,y^{22}\,d[34])$

d4

In[36]:=
```
M11integrand = Series[M2[5], {y, 0, 8}] // Simplify // Normal;
M11parts = CoefficientList[M11integrand, x];
res =
  MapIndexed[#1 /. {g1[x] → g1[First[#2 - 1], nMax], g2[x] → g2[First[#2 - 1], nMax]} &,
   M11parts] // Simplify;
Total[res] // Simplify
```

Out[39]=

$$d[4] + y^{10} \left(1001\, d[14] + 11\,375\, y^2\, d[16] + 71\,500\, y^4\, d[18] + 327\,250\, y^6\, d[20] + 1\,215\,500\, y^8\, d[22] + 3\,879\,876\, y^{10}\, d[24] + 11\,022\,375\, y^{12}\, d[26] + 28\,528\,500\, y^{14}\, d[28] + 68\,392\,870\, y^{16}\, d[30] + 153\,893\,350\, y^{18}\, d[32] + 323\,323\,000\, y^{20}\, d[34] + 743\,642\,900\, y^{22}\, d[36]\right)$$

Check for a' s

a0

In[40]:=
```
nMax = 12;
M11integrand = Series[M1[1], {y, 0, 8}] // Simplify // Normal;
M11parts = CoefficientList[M11integrand, x];
res =
  MapIndexed[#1 /. {g1[x] → g1[First[#2 - 1], nMax], g2[x] → g2[First[#2 - 1], nMax]} &,
   M11parts] // Simplify;
Total[res] // Simplify
```

Out[44]=

$$a[0] + \frac{4}{11} y^{10} (3\, a[10] + 140\, d[10]) + \frac{385}{169} y^{12} (3\, a[12] + 176\, d[12]) + \frac{88}{45} y^{14} (13\, a[14] + 924\, d[14]) + \frac{1001}{289} y^{16} (21\, a[16] + 1760\, d[16]) + \frac{12\,740}{361} y^{18} (5\, a[18] + 484\, d[18]) + \frac{1144}{3} y^{20} (a[20] + 110\, d[20]) + \frac{7840}{529} y^{22} (51\, a[22] + 6292\, d[22]) + \frac{19\,448}{125} y^{24} (9\, a[24] + 1232\, d[24]) + \frac{182}{729} y^{26} (9839\, a[26] + 1\,481\,036\, d[26]) + \frac{14\,560\, y^{28} (714\, a[28] + 117\,269\, d[28])}{2523} + \frac{12\,376}{961} y^{30} (513\, a[30] + 91\,540\, d[30]) + \frac{587\,860}{363} y^{32} (7\, a[32] + 1328\, d[32])$$

a2

In[45]:=
```
nMax = 12;
M11integrand = Series[M1[3], {y, 0, 8}] // Simplify // Normal;
M11parts = CoefficientList[M11integrand, x];
res =
  MapIndexed[#1 /. {g1[x] → g1[First[#2 - 1], nMax], g2[x] → g2[First[#2 - 1], nMax]} &,
    M11parts] // Simplify;
Total[res] // Simplify
```

Out[49]=

$$a[2] + \frac{396}{13} y^{10} (a[12] + 12\, d[12]) + \frac{858}{25} y^{12} (7\, a[14] + 104\, d[14]) + \frac{1404}{289} y^{14} (225\, a[16] + 4004\, d[16]) + \frac{270270}{361} y^{16} (5\, a[18] + 104\, d[18]) + \frac{880}{7} y^{18} (85\, a[20] + 2028\, d[20]) + \frac{525096}{529} y^{20} (27\, a[22] + 728\, d[22]) + \frac{2004912}{625} y^{22} (19\, a[24] + 572\, d[24]) + \frac{46189}{9} y^{24} (25\, a[26] + 832\, d[26]) + \frac{1560}{841} y^{26} (136731\, a[28] + 4988356\, d[28]) + \frac{23868\, y^{28} (95855\, a[30] + 3805368\, d[30])}{4805} + \frac{705432}{121} y^{30} (145\, a[32] + 6244\, d[32]) + \frac{831402}{35} y^{32} (69\, a[34] + 3152\, d[34])$$

d2: Oscars method

Check

In[50]:=
$$f1 = \left(2\frac{\xi}{x} - 3\left(1 - \frac{\xi^2 y^2}{2}\right)\right);$$
$$fT = 3\left(1 - \frac{\xi^2 y^2}{2}\right);$$

```
gTterm = Series[ξ^2 fT (g1[x] + g2[x]), {y, 0, nMax}];
gTpart = CoefficientList[gTterm, x];
g1term = Series[ξ^2 f1 (g1[x]), {y, 0, nMax}];
g1part = CoefficientList[g1term, x];
res =
  MapIndexed[#1 /. {g1[x] → g1[First[#2 - 1], nMax], g2[x] → g2[First[#2 - 1], nMax]} &,
   gTpart] // Simplify;
gTtotal = Total[res] // Simplify;
res =
  MapIndexed[#1 /. {g1[x] → g1[First[#2 - 1], nMax], g2[x] → g2[First[#2 - 1], nMax]} &,
   g1part] // Simplify;
g1total = Total[res] // Simplify;
Method1 = gTtotal + g1total // Simplify
```

Out[60]=
$$2\left(d[2] + 2 y^{14}\left(1716\, d[16] + 23\,205\, y^2 d[18] + 170\,170\, y^4 d[20] + 895\,356\, y^6 d[22] + 3\,779\,100\, y^8 d[24] + 13\,579\,566\, y^{10} d[26] + 43\,088\,955\, y^{12} d[28] + 123\,735\,690\, y^{14} d[30] + 327\,085\,304\, y^{16} d[32] + 809\,924\,115\, y^{18} d[34] + 1\,829\,361\,534\, y^{20} d[36] + 4\,833\,678\,850\, y^{22} d[38]\right)\right)$$

It works as expected.

Leaving out twist-3 target mass effects to g1

In[61]:=
```
gTterm = Series[ξ^2 fT (g1[x] + g2[x]), {y, 0, nMax}];
gTpart = CoefficientList[gTterm, x];
g1term = Series[ξ^2 f1 (g1[x]), {y, 0, nMax}];
g1part = CoefficientList[g1term, x];
res =
  MapIndexed[#1 /. {g1[x] → g1[First[#2 - 1], nMax], g2[x] → g2[First[#2 - 1], nMax]} &,
    gTpart] // Simplify;
gTtotal = Total[res] // Simplify;
(* note that in the line below g1tw2 is used *)
res = MapIndexed[
    #1 /. {g1[x] → g1tw2[First[#2 - 1], nMax], g2[x] → g2[First[#2 - 1], nMax]} &,
    g1part] // Simplify;
g1total = Total[res] // Simplify;
Method2 = gTtotal + g1total // Simplify
gTtotal + g1total // Simplify // N
```

Out[69]= $2\, d[2] + \frac{48}{25} y^2 d[4] + \frac{60}{49} y^4 d[6] + \frac{44}{27} y^6 d[8] + \frac{168}{121} y^8 d[10] + \frac{264}{169} y^{10} d[12] + \frac{36}{25} y^{12} d[14] + \frac{1\,984\,140}{289} y^{14} d[16] + \frac{34\,033\,644}{361} y^{16} d[18] + \frac{34\,237\,984}{49} y^{18} d[20] + \frac{1\,959\,776\,148}{529} y^{20} d[22] + \frac{9\,829\,064\,076}{625} y^{22} d[24] + \frac{13\,792\,782\,536}{243} y^{24} d[26] + \frac{151\,995\,455\,196}{841} y^{26} d[28] + \frac{500\,150\,630\,304}{961} y^{28} d[30] + \frac{166\,839\,740\,392}{121} y^{30} d[32] + \frac{4\,191\,262\,099\,596\, y^{32} d[34]}{1225} + \frac{10\,586\,151\,874\,584\, y^{34} d[36]}{1369} + 20\,511\,469\,000\, y^{36} d[38]$

Out[70]= $2.\, d[2.] + 1.92\, y^2 d[4.] + 1.22449\, y^4 d[6.] + 1.62963\, y^6 d[8.] + 1.38843\, y^8 d[10.] + 1.56213\, y^{10} d[12.] + 1.44\, y^{12} d[14.] + 6865.54\, y^{14} d[16.] + 94\,276.\, y^{16} d[18.] + 698\,734.\, y^{18} d[20.] + 3.70468\times10^6\, y^{20} d[22.] + 1.57265\times10^7\, y^{22} d[24.] + 5.67604\times10^7\, y^{24} d[26.] + 1.80732\times10^8\, y^{26} d[28.] + 5.20448\times10^8\, y^{28} d[30.] + 1.37884\times10^9\, y^{30} d[32.] + 3.42144\times10^9\, y^{32} d[34.] + 7.73276\times10^9\, y^{34} d[36.] + 2.05115\times10^{10}\, y^{36} d[38.]$

Check against 2g1+3g2 leaving g1 with twist-2 TMCs only

In[71]:=
```
res = MapIndexed[
    #1 /. {g1[x] → g1tw2[First[#2 - 1], nMax], g2[x] → g2[First[#2 - 1], nMax]} &,
    Iparts] // Simplify;
Method3 = Total[res] // Simplify
Total[res] // Simplify // N
```

Out[72]=
$$2\,d[2] + \frac{54}{25} y^2 d[4] + \frac{108}{49} y^4 d[6] + \frac{20}{9} y^6 d[8] + \frac{270}{121} y^8 d[10] + \frac{378}{169} y^{10} d[12] + \frac{56}{25} y^{12} d[14] + \frac{648}{289} y^{14} d[16] + \frac{810}{361} y^{16} d[18] + \frac{110}{49} y^{18} d[20] + \frac{1188}{529} y^{20} d[22] + \frac{1404}{625} y^{22} d[24] + \frac{182}{81} y^{24} d[26]$$

Out[73]=
$$2.\,d[2.] + 2.16\, y^2 d[4.] + 2.20408\, y^4 d[6.] + 2.22222\, y^6 d[8.] + 2.2314\, y^8 d[10.] + 2.23669\, y^{10} d[12.] + 2.24\, y^{12} d[14.] + 2.24221\, y^{14} d[16.] + 2.24377\, y^{16} d[18.] + 2.2449\, y^{18} d[20.] + 2.24575\, y^{20} d[22.] + 2.2464\, y^{22} d[24.] + 2.24691\, y^{24} d[26.]$$

This is actually BETTER than if g1 included the twist - 3 TM effects!
Compare to the following note the factor of 6 below and 54/25 above.

In[74]:=
```
res =
  MapIndexed[#1 /. {g1[x] → g1[First[#2 - 1], nMax], g2[x] → g2[First[#2 - 1], nMax]} &,
    Iparts] // Simplify;
Method0 = Total[res] // Simplify // Expand
```

Out[75]=
$$2\,d[2] + 6\,y^2 d[4] + 12\,y^4 d[6] + 20\,y^6 d[8] + 30\,y^8 d[10] + 42\,y^{10} d[12] + 56\,y^{12} d[14] + 72\,y^{14} d[16] + 90\,y^{16} d[18] + 110\,y^{18} d[20] + 132\,y^{20} d[22] + 156\,y^{22} d[24] + 182\,y^{24} d[26]$$

3 gT - g1(tw2)

In[76]:=
```
gTterm = Series[x^2 3 (g1[x] + g2[x]), {y, 0, nMax}];
gTpart = CoefficientList[gTterm, x];
g1term = Series[-x^2 (g1[x]), {y, 0, nMax}];
g1part = CoefficientList[g1term, x];
res = MapIndexed[
    #1 /. {g1[x] → g1tw2[First[#2 - 1], nMax], g2[x] → g2[First[#2 - 1], nMax]} &,
    gTpart] // Simplify;
gTtotal = Total[res] // Simplify;
res =
  MapIndexed[#1 /. {g1[x] → g1[First[#2 - 1], nMax], g2[x] → g2[First[#2 - 1], nMax]} &,
    g1part] // Simplify;
g1total = Total[res] // Simplify;
Method4 = gTtotal + g1total // Simplify
gTtotal + g1total // Simplify // N
```

Out[84]=

$$2\,d[2] + \frac{6}{25} y^2 d[4] - \frac{132}{49} y^4 d[6] - \frac{20}{3} y^6 d[8] - \frac{1410}{121} y^8 d[10] - \frac{2982}{169} y^{10} d[12] - \frac{616}{25} y^{12} d[14] - \frac{9432}{289} y^{14} d[16] - \frac{15\,030}{361} y^{16} d[18] - \frac{2530}{49} y^{18} d[20] - \frac{33\,132}{529} y^{20} d[22] - \frac{46\,644}{625} y^{22} d[24] - \frac{2366}{27} y^{24} d[26]$$

Out[85]=

$$2.\,d[2.] + 0.24\, y^2 d[4.] - 2.69388\, y^4 d[6.] - 6.66667\, y^6 d[8.] - 11.6529\, y^8 d[10.] - 17.645\, y^{10} d[12.] - 24.64\, y^{12} d[14.] - 32.6367\, y^{14} d[16.] - 41.6343\, y^{16} d[18.] - 51.6327\, y^{18} d[20.] - 62.6314\, y^{20} d[22.] - 74.6304\, y^{22} d[24.] - 87.6296\, y^{24} d[26.]$$

Check

In[86]:=
```
M = 0.938;
Q2 = 1.0;
y = Sqrt[M^2 / Q2]
Plot[{f1, fT}, {x, 0.01, 0.99}]
Clear[Q2, M, y]
```

Out[88]= 0.938

Out[89]=



Comparing Methods

In[91]:=
```
dropAbove[a_, n_] :=
 Total[MapIndexed[#1 * y^First[#2 - 1] &, Take[CoefficientList[a, y], n]]]
highestN = 6;
dropAbove[Method0, highestN]
dropAbove[Method1, highestN]
dropAbove[Method2, highestN]
dropAbove[Method3, highestN]
dropAbove[Method4, highestN]
```

Out[93]= $2\,d[2] + 6\,y^2\,d[4] + 12\,y^4\,d[6]$

Out[94]= $2\,d[2]$

Out[95]= $2\,d[2] + \frac{48}{25}\,y^2\,d[4] + \frac{60}{49}\,y^4\,d[6]$

Out[96]= $2\,d[2] + \frac{54}{25}\,y^2\,d[4] + \frac{108}{49}\,y^4\,d[6]$

Out[97]= $2\,d[2] + \frac{6}{25}\,y^2\,d[4] - \frac{132}{49}\,y^4\,d[6]$

In[98]:=
```
{ICN == dropAbove[Method0, highestN],
 IOscar == dropAbove[Method2, highestN],
 I3gTMinusg1 == dropAbove[Method4, highestN]}
```

Out[98]= $\left\{ICN == 2\, d[2] + 6\, y^2\, d[4] + 12\, y^4\, d[6],\ IOscar == 2\, d[2] + \frac{48}{25}\, y^2\, d[4] + \frac{60}{49}\, y^4\, d[6],\ I3gTMinusg1 == 2\, d[2] + \frac{6}{25}\, y^2\, d[4] - \frac{132}{49}\, y^4\, d[6]\right\}$

In[99]:=
```
res = Solve[{I0 == dropAbove[Method0, highestN],
      I2 == dropAbove[Method2, highestN],
      I4 == dropAbove[Method4, highestN]}, {d[2], d[4], d[6]}] // Simplify //
   Flatten;
res // Expand // TableForm
{d[2], d[4], d[6]} /. res // Expand // TableForm
({d[2], d[4], d[6]} /. res) // Expand // TableForm // N /. {y → 1}
res /. {y → 1} // Expand // TableForm
```

Out[100]//TableForm=

$d[2] \to \frac{31\, I0}{24} - \frac{9\, I2}{2} + \frac{89\, I4}{24}$

$d[4] \to -\frac{50\, I0}{27\, y^2} + \frac{125\, I2}{18\, y^2} - \frac{275\, I4}{54\, y^2}$

$d[6] \to \frac{343\, I0}{432\, y^4} - \frac{49\, I2}{18\, y^4} + \frac{833\, I4}{432\, y^4}$

Out[101]//TableForm=

$\frac{31\, I0}{24} - \frac{9\, I2}{2} + \frac{89\, I4}{24}$

$-\frac{50\, I0}{27\, y^2} + \frac{125\, I2}{18\, y^2} - \frac{275\, I4}{54\, y^2}$

$\frac{343\, I0}{432\, y^4} - \frac{49\, I2}{18\, y^4} + \frac{833\, I4}{432\, y^4}$

Out[102]//TableForm=

$1.29167\, I0 - 4.5\, I2 + 3.70833\, I4$

$-\frac{1.85185\, I0}{y^2} + \frac{6.94444\, I2}{y^2} - \frac{5.09259\, I4}{y^2}$

$\frac{0.793981\, I0}{y^4} - \frac{2.72222\, I2}{y^4} + \frac{1.92824\, I4}{y^4}$

Out[103]//TableForm=

$d[2] \to \frac{31\, I0}{24} - \frac{9\, I2}{2} + \frac{89\, I4}{24}$

$d[4] \to -\frac{50\, I0}{27} + \frac{125\, I2}{18} - \frac{275\, I4}{54}$

$d[6] \to \frac{343\, I0}{432} - \frac{49\, I2}{18} + \frac{833\, I4}{432}$